Appendix S1. BUGS code for the N-mixture model used to evaluate the response of Aspidoscelis sexlineata, Sceloporus consobrinus, and Scincella lateralis to high-severity wildfires on the Griffith League Ranch, Bastrop County, Texas, USA, from spring 2008 – spring 2013.

model {

    ##################################################

    ##Priors 

    ###################################################

    #detection

    alpha0~dnorm(0,0.37)

    for(k in 1:5){

      alpha[k]~dnorm(0,0.37)

    }    

    #abundance

    beta0~dnorm(0,0.001)

    for(j in 1:3){

      beta[j]~dnorm(0,0.001)

    }

    ##################################################

    ##Ecological model for true abundance 

    ################################################### 

    for(t in 1:T){

      for(i in 1:M){

        N[i,t]~dpois(lambda[i,t])

        log.lambda.1[i,t]<-beta0+

                           beta[1]*Hab[i]+

                           beta[2]*Burn[i,t]+

                           beta[3]*Year[t]

        log.lam[i,t]<-min(250,max(-250,log.lambda.1[i,t])) #Stabilize log

        lambda[i,t]<-exp(log.lam[i,t])

        ##################################################

        ##Observation model for replicated counts

        ###################################################  

        for(j in 1:J){

          C[i,j,t]~dbin(p[i,j,t],N[i,t])

          logit(p0[i,j,t])<-alpha0+

                            alpha[1]*Dates[j,t]+

                            alpha[2]*pow(Dates[j,t],2)+

                            alpha[3]*Temp[j,t]+

                            alpha[4]*Prec[j,t]+

                            alpha[5]*Burn[i,t]

          p[i,j,t]<-max(0.001,min(0.999, p0[i,j,t])) #stabilize logit

          ##################################################

          ##Assess model fit using Chi-squared discrepancy

          ##################################################

          eval[i,j,t]<-p[i,j,t]*N[i,t] # Expected values

          E[i,j,t]<-pow((C[i,j,t]-eval[i,j,t]),2)/(eval[i,j,t]+0.5)

          y.new[i,j,t]~dbin(p[i,j,t],N[i,t])

          E.new[i,j,t]<-pow((y.new[i,j,t]-eval[i,j,t]),2)/(eval[i,j,t]+0.5)

        }

      }

    }

    #fill in missing values for precipitation

    for(ed in 1:J){

      for(ld in 1:T){

        Prec[ed,ld]~dnorm(mu.prec,tau.prec)

      }

    }

    mu.prec~dnorm(0,0.01)

    tau.prec<-pow(sd.prec,-2)

    sd.prec~dunif(0,100)

    #######################

    ## Derived quantities

    #######################

    for(t in 1:T){

       totalN[t]<-sum(N[,t])

    }

    logit(mean.p)<-alpha0

    fit<-sum(E[,,])         # Fit stats actual data set

    fit.new<-sum(E.new[,,]) # Fit stats ‘ideal’ data set

    } 
